

ぺるけさんの全段差動プッシュプルアンプを シミュレートする (ベーシックアンプ篇)

Ayumi's Lab.

2002年12月16日

Updated 2002年12月18日

ぺるけさんが主催している“Building My Very First Tube Amp 講座” (<http://www2.famille.ne.jp/~teddy/myamp/myamp.htm>) のベーシックアンプの回路をシミュレートしてみました。このアンプをお作りになったかたは、ぜひこのシミュレーションの結果と比較してみてください。

シミュレーションなので、以下のような制約があることをお含みおきください。

- 真空管は完全なマッチドペア
- 浮遊容量は入れていない(電極間容量は入っている)
- 電源には内部抵抗がない(全段差動には影響はない)

1 全体の構成

初段は 6SC7，出力段は 6AH4 とします。

初段のプレート供給電圧は 255 V，負荷抵抗は 220 k Ω とします。定電流値は 1.1 mA とします。このとき、カソードの電位は、1.476 V になります。

出力段のプレート供給電圧を 266 V，カソード電流は $I_k = 62.5$ mA とします。カソードから定電流源には、プレート電流確認用の 4.7 Ω が入っていますが、この抵抗があると解析が難しくなるため、取り払っています。カソードの電位は $E_k = 21.3$ V となります。出力トランスのインピーダンスは 8 k Ω としました。OPT は、タンゴまたはノグチを推奨していますが、詳細なデータが手に入らないのでタムラ製作所の F685 を使用しました。OPT による電圧降下は 6.31 V です(片側の巻線抵抗は 202 Ω です)。したがって、無信号時のプレート電圧は $266 - 6.31 - 21.3 = 238.4$ V となります。プレート損失は、 $238.4 \times 0.03125 = 7.45$ W です。

NFB は、ぺるけさんの定数では 2.4 k Ω と 100 Ω で分圧しています。出力トランスの 2 次側 8 Ω 端子までのトータルゲインが 7.74 倍なので、負帰還量 F は、

$$F = 1 + A\beta = 1 + 7.74 \frac{100}{100 + 2400} = 1.3096$$

より 2.35 dB となります。

2 ロードライン

電圧増幅段のロードラインを図 1 に示します。青色の線が $220\text{k}\Omega$ の直流ロードラインで、緑色の線が $220//470 = 150\text{k}\Omega$ の交流ロードラインです。フルパワー時、入力の尖頭値が 0.9V を超えるので、初速度電流の影響を避けるには、もう少しバイアスを深くしたほうがよいのではと思います。




図 1: 電圧増幅段のロードライン

出力段のロードラインを図 2 に示します。緑色の線がプレート電圧とプレート電流の関係を表わしており、この場合、入力は -24V から 24V を加えています。差動出力段ではカソードの電圧が変動するため、ロードラインは弓なりになります。青色の線は $4\text{k}\Omega$ のシングル動作のロードラインです。茶色の線は、プレート電流に対応したカソード電圧です。カソードの電圧は、 21.3V から 26.1V まで変動します。

出力段の伝達特性を図 3 に示します。横軸は上側の球のグリッド電圧で、下側の球には位相が逆の電圧が加わっています。赤い色の線が上側の球のプレート電圧で、青い色の線が下側の球のプレート電流です。

最大出力時のカソード電圧を求めるには、片方の真空管に設定した電流がすべて流れ、もう一方はカットオフした状態を考えます。この状態の時、電流が流れている真空管の(対カソード)グリッド電圧は、ロードラインと共通カソード電流(ここでは 62.5mA) が交わる点から求められ、ここでは約 -2V です。カットオフしている真空管のグリッド電圧は、約 -50V です。このとき、両グリッド間で $50 - 2 = 48\text{V}$ の入力がかかっています。それぞれの真空管には $48/2 = 24\text{V}$ の入力がかかります。しかし、電流が流れる方の真空管のグリッド電圧は、動作点の -21.3V から -2V に 19.3V 変化しているはずですし、カットオフしている方のグリッド電圧は、 -21.3V から -50V に -28.7V 変化しているはずで、入力信号 ($\pm 24\text{V}$) とグリッド電圧の差を調整するようにカソードの電圧が上昇します。すなわち、 $24 - 19.3 = 4.7\text{V}$ または $28.7 - 24 = 4.7\text{V}$ だけカソード電圧が上




図 2: 出力段のロードライン




図 3: 出力段の伝達特性

昇します。ここでは、カソード電圧の上昇によってプレート電圧が下がる効果を考えていませんが、カソード電圧の変化は数ボルトなので、ここで説明した方法である程度の推定はできるでしょう。

3 シミュレーションプログラム

Spice3f5用のシミュレーションプログラムを以下に示します。グラフ表示等のコマンドは、使用するSPICEに合わせて適宜変更してください。

6SC7-6AH4.cir

```
1 Mr. Perche's 6AH4 differential basic amplifier
2 .INCLUDE 6SC7.lib
3 .INCLUDE 6AH4.lib
4 .INCLUDE F685.lib
5 *
6 Rg1 1 0 100k
7 Rg2 2 0 100
8 X1 4 1 3 6SC7
9 X2 5 2 3 6SC7
10 Ik1 3 0 1.1mA
11 Rp1 4 6 220k
12 Rp2 5 6 220k
13 Vbb1 6 0 255V
14 Cg1 4 11 0.22u
15 Cg2 5 12 0.22u
16 Rg3 11 0 470k
17 Rg4 12 0 470k
18 X3 14 11 13 6AH4
19 X4 15 12 16 6AH4
20 Rk1 13 17 1m
21 Rk2 16 17 1m
22 Ik2 17 0 62.5mA
23 XOPT 24 20 20 25 30 0 F685
24 RL 30 0 8ohm
25 Vbb2 20 0 266V
26 Vip1 24 14 DC 0V
27 Vip2 25 15 DC 0V
28 Vin 1 0 DC 0V AC 1V SIN(0 1.05V 1kHz)
29 **Vin 1 0 DC 0V AC 1V PULSE(-0.918V 0.918V 0 0.1m 0.1m 5m 10m)
30 *Rnfb 30 2 2.4k
31 .NODESET V(3)=1.27V V(13)=20V V(16)=20V
32 *
33 .control
34 set width=240
35 *op
36 *print v(4) v(1) v(3) v(4,3) v(1,3)
37 *print v(14) v(11) v(13) v(14,13) v(11,13)
38 *print v(20,24)
39 *ac dec 10 1k 1k
40 *print abs(v(4)) abs(v(30))
41 *ac dec 10 1 1Meg
42 *plot db(v(30))
43 *plot ph(v(30))
44 *print db(v(30)) ph(v(30))
45 *tran 0.005ms 2ms 0 0.0025ms
46 *plot v(11,13)
47 *plot i(vip1)
48 *fourier 1k v(30)
49 *linearize v(4) v(5) v(3) v(11) v(12) v(13) v(16) v(30) i(vip1) i(vip2)
50 *print v(4) v(5) v(3) v(11) v(12) v(13) v(16) v(30) i(vip1) i(vip2)
51 *print sqrt(mean(v(30)^2)) mean(v(30)^2)/8
52 .endc
53 .END
```

6SC7.lib

```
1 *
2 * GENERIC: 6SC7
3 *
4 .SUBCKT 6SC7 A G K
5 BGG GG 0 V=V(G,K)+0.42675
6 BEP EP 0 V=URAMP(V(A,K))+1e-10
7 BSTM STM 0 V=URAMP(V(GG)+V(EP)/60.324)+1e-10
8 BM1 M1 0 V=(0.00279159206949141*(URAMP(V(EP)-1e-10)+1e-10))^-0.303751803751804
9 BM2 M2 0 V=(0.8316*V(STM))^1.80375180375180
10 BM M 0 V=V(M1)*V(M2)
11 BSTP STP 0 V=URAMP(V(GG)+V(EP)/72.5396825396825)+1e-10
12 BP P 0 V=1.62432850439450*V(STP)^1.5
13 BIK IK 0 V=U(V(GG))*V(P)+(1-U(V(GG)))*V(M)
14 BLIM LI 0 V=0.88*V(EP)^1.5
15 BEG EG 0 V=URAMP(V(G,K))+1e-10
16 BIG IG 0 V=0.88*(V(EG)/(V(EP)+V(EG)))*1.2+0.4)*V(EG)^1.5
17 BIP IP 0 V=(V(IK,IG)-URAMP(V(IK,IG)-V(LI)))
18 BIGK G K I=0.0004426*V(IG)
19 BIAK A K I=0.0004426*V(IP)
20 * CAPS
21 CAK A K 3e-12
22 CGK G K 2e-12
23 CGA G A 2e-12
24 .ENDS
```

6AH4.lib

```
1 *
2 * GENERIC: 6AH4
3 *
4 .SUBCKT 6AH4 A G K
5 BGG GG 0 V=V(G,K)+1.2375
6 BEP EP 0 V=URAMP(V(A,K))+1e-10
7 BSTM STM 0 V=URAMP(V(GG)+V(EP)/7.262)+1e-10
8 BM1 M1 0 V=(0.040277471770862*(URAMP(V(EP)-1e-10)+1e-10))^-0.620126359531028
9 BM2 M2 0 V=(0.707505*V(STM))^2.12012635953103
10 BM M 0 V=V(M1)*V(M2)
11 BSTP STP 0 V=URAMP(V(GG)+V(EP)/10.2642384152762)+1e-10
12 BP P 0 V=1.72936766471006*V(STP)^1.5
13 BIK IK 0 V=U(V(GG))*V(P)+(1-U(V(GG)))*V(M)
14 BLIM LI 0 V=0.88*V(EP)^1.5
15 BEG EG 0 V=URAMP(V(G,K))+1e-10
16 BIG IG 0 V=0.88*(V(EG)/(V(EP)+V(EG)))*1.2+0.4)*V(EG)^1.5
17 BIP IP 0 V=(V(IK,IG)-URAMP(V(IK,IG)-V(LI)))
18 BIGK G K I=0.0011961*V(IG)
19 BIAK A K I=0.0011961*V(IP)
20 * CAPS
21 CAK A K 1.7e-12
22 CGK G K 7e-12
23 CGA G A 4.4e-12
24 .ENDS
```

F685.lib

```
1 *
2 * TAMURA OPT F685
3 *
4 .SUBCKT F685 P1 B1 B2 P2 S8 S0
5 * 1次インダクタンス (p-p 8kohm 200H)
6 L11 P1 12 50H
7 L12 13 P2 50H
8 * 1次巻線直流抵抗
9 R11 12 B1 202
10 R12 B2 13 202
11 C11 P1 B1 1150p
12 C12 P2 B2 1150p
```

```
13 * 2次インダクタンス (8ohm)
14 L13 S8 16 0.2206951H
15 * 2次巻線抵抗
16 R13 S0 16 0.382
17 K1 L11 L13 0.99998
18 K2 L12 L13 0.99998
19 K3 L11 L12 0.99998
20 .ENDS
```

4 無帰還時の各部の波形

$V_i = 0.74 \text{ V}$ (実効値) 1 kHz の正弦波を入力した時の各部の波形を示します。このときの出力は $P_o = 3.66 \text{ W}$, 歪率は $\text{THD} = 2.77 \%$ です。グラフの赤い線は V_{1a} または V_2 , グラフの青い線は V_{1b} または V_3 の電圧を表わしています。

図 4は, V_{1a}, V_{1b} のプレートの電圧です。図 5は, V_{1a}, V_{1b} のカソードの電圧です。入力のほぼ半分の振幅の波形がカソードに現われます。




図 4: 6SC7 のプレートの電圧




図 5: 6SC7 のカソードの電圧

図 6は, V_2, V_3 のグリッドの電圧です。これは, V_{1a}, V_{1b} のプレートの電圧を 0 V が中心となるようにシフトした波形になっています。図 7は, V_2, V_3 のカソードの電圧です。出力段は平衡ドライブされているので, カソードに現われる信号の周波数は, 入力信号の 2 倍になっています。ま

た、カソード電圧は無信号時に最低となり、入力加わるとそれ以上に上昇します。この回路の場合、カソード電圧が最低でも約 21 V あるので、トランジスタによる定電流回路を使えば、定電流回路の帰路はアースでよいことになります。




図 6: 6AH4 のグリッドの電圧




図 7: 6AH4 のカソードの電圧

最大出力時には、カソード電圧が約 4 V 上昇します。正味のグリッド電圧はその分差し引かれるので、グリッド電圧が 0 V になるまでフルスイングするのに必要な入力は波高値で

$$\text{無信号時のグリッド電圧 (約 21 V) + 4} \approx 25 \text{ V}$$

となります¹。実効値では約 17.7 V です。この正味のグリッド電圧の波形は、図 8 のようになります。かなり歪んだ波形ですが、これにより 6AH4 のリニアリティの悪さが補正され、出力にはかな

¹この出力段の定数では、グリッド電圧が 0 になる前に反対側の球がカットオフしてしまうので、これだけの入力を与えると正弦波の頭がつぶれてしまいます。

りきれいな正弦波が現われます。図 9は、出力管のプレート電流の波形です。図 10は、 8Ω 出力端子の波形です。




図 8: 6AH4 の正味グリッド電圧




図 9: 6AH4 のプレート電流

一方のグリッドに $+23\text{V}$ 、もう一方のグリッドに -23V を加えた時に、各部の電圧や電流がどう変化するかを表わしているのが図 11です。

ここで、この図の数値とロードラインの関係を見てみましょう。無信号時には、 $V3$ 、 $V4$ とも図 2の“O”点にあります。 $V3$ のグリッドに $+23\text{V}$ 、 $V4$ のグリッドに -23V を加えると、 $V3$ は“A”点に移動し、 $V4$ は“B”点に移動します。これらの点は緑色の線(差動出力段のロードライン)に乗っていることがわかります。




図 10: 出力電圧




図 11: 無信号時と最大出力時の各部の電圧

5 小信号解析

初段のゲインは21.9倍(26.8dB), トータルゲイン(8Ω)は7.74倍(17.8dB)でした。ゲインの周波数特性を図12に, 位相の周波数特性を図13に示します。赤い線は無帰還時の特性で, 青い線は2.5dBの負帰還をかけたときの特性です。




図 12: ゲインの周波数特性




図 13: 位相の周波数特性

6 大振幅特性

入出力特性を図14に, 出力対歪率特性を図15に示します。赤い線は無帰還時の特性で, 青い線は2.35dBの負帰還をかけたときの特性です。




图 14: 入出力特性




图 15: 出力对歪率特性